



PUBLIC

How To Create an Enrichment Spot with User Interaction in Central Governance

Applicable Releases:

From EHP6 for SAP ERP 6.0 and from S/4HANA 1511

Version 1.1

January 2024

Document History

Document Version	Description
1.0	First official release of this guide (2012)
1.1	Adapted new template (2024)

1	BUSINESS SCENARIO.....	4
1.1	Scenario <i>Create Airline</i>	4
1.2	Scenario <i>Process Airline</i>	6
2	CREATE CHANGE REQUEST TYPE.....	9
3	ASSIGN PROCESSOR TO WORKFLOW STEP NUMBER	9
4	CREATE ENRICHMENT ADAPTER CLASS.....	10
5	CREATE ENRICHMENT FEEDER CLASS.....	11
6	CREATE POPUP WICH DISPLAYS A TABLE OF URLS	14
7	DEFINE ENRICHMENT SPOT	18
8	CONFIGURE PROPERTIES OF CHANGE REQUEST STEP	19
9	ADDITIONAL INFORMATION	21
9.1	Further Reading	21
9.1.1	Information on SAP MDG on SAP S/4HANA	21
9.1.2	SAP Roadmap Explorer	21
9.1.3	Related Information	21
9.2	SAP Notes.....	21

1 Business Scenario

SAP Master Data Governance (MDG) provides business processes to find, create, change, and mark master data for deletion. It supports the governance of master data in a central hub and the distribution to connected operational and business intelligence systems.

The processes are workflow-driven and can include several approval and revision phases, and the collaboration of all users participating in the master data maintenance.

MDG offers change request (CR)-based processing of master data with integrated workflow, staging, approval, activation, and distribution.


This guide describes how to create an enrichment spot implementation with user interaction in Master Data Governance. The implementation is called when executing a consistency check in the Single Processing UI (WebDynpro Application `usmd_entity_value2`).

The business scenario comprises the creation of a change request for a single airline and the processing of the change request. When creating a change request or processing a change request a consistency check can be executed in the Single Processing UI.

The consistency check processing shall be enhanced by a customer specific implementation (enrichment). The customer specific implementation shall be executed when a user clicks on the 'Check' button in the Single Processing UI.

1.1 Scenario *Create Airline*

Call the Single Processing UI to create an airline.



Enter an airline and select a change request type which refers to an enrichment implementation (the definition of the change request type is described in the following chapters).

Create Change Request: Airline MH9

Submit | Check | Save Draft

Entity Type Airline

▼ Change Request

General Data | Notes | Attachments

Description: * testmh1338

Priority: ▼

Main Data | Attachments

Airline Details

Airline: * mh9

Airline URL: * http://www.ford.com

Airline local currency: EUR

Now enter a description, name of airline, airline URL and currency. Then click on the 'Check' button to execute a consistency check. During the check processing the customer specific enrichment implementation is called which displays a popup containing an URL for an airline ID.

Create Change Request: Airline MH9

Submit | Check | Save Draft

Entity Type Airline

▼ Change Request

General Data | Notes | Attachments

Description: * testmh1338

Priority: ▼

Main Data | Attachments

Airline Details

Airline: * mh9

Airline URL: * http://www.ford.com

Airline local currency: EUR

URL Selection

Client	Airline	URL
201	MH9	www.sap.com

OK Cancel

When you click on the 'OK' button the selected URL 'www.sap.com' in the popup is transferred to the field Airline URL on the UI.

Create Change Request: Airline MH9

Submit | Check | Save Draft

☒ Airline MH9 URL overwritten

Entity Type: Airline

▼ Change Request

General Data | Notes | Attachments

Description: testmh1338

Priority: ▼

Main Data | Attachments

Airline Details

Airline: mh9

Airline URL: http://www.sap.com

Airline local currency: EUR

Remark: When you press the 'Cancel' button in the popup the value in field Airline URL won't be overwritten. Now press the 'Submit' button to create the change request.

Create Change Request: Airline MH9

☒ Change request 625 created

What would you like to do next?

[Create Another Change Request](#)

[Show My Change Requests](#)

1.2 Scenario *Process Airline*

Display the previously created change request.

Process Change Request: Airline MH9

Save Finalize Processing Send for Revision Close Read Only Check Show ▾

Change Request 625 Type Process Airline Status Changes to Be Executed

▼ Change Request

General Data Notes Attachments

Processing: You are the processor of the change request

Description: testmh1338

Priority: ▾

Created On: 23.11.2012 13:50:55

Changed On:

Finalized On:

Main Data Attachments

Airline Details

Airline: mh9

Airline URL: http://www.sap.com

Airline local currency: EUR

Now change the value of field Airline URL to 'www.bmw.com'.

Process Change Request: Airline MH9

Save Finalize Processing Send for Revision Close Read Only Check Show ▾

Change Request 625 Type Process Airline Status Changes to Be Executed

▼ Change Request

General Data Notes Attachments

Processing: You are the processor of the change request

Description: testmh1338

Priority: ▾

Created On: 23.11.2012 13:50:55

Changed On:

Finalized On:

Main Data Attachments

Airline Details

Airline: mh9

Airline URL: www.bmw.com

Airline local currency: EUR

Then click on the 'Check' button to execute a consistency check. During the check processing the customer specific enrichment implementation is called which displays a popup containing an URL for an airline ID.

Process Change Request: Airline MH9

Save Finalize Processing Send for Revision Close | Read Only | Check Show ▾ Run Validation Validation Log Related Services ▾

☒ Airline MH9 URL overwritten

Change Request 625 Type Process Airline Status Changes to Be Executed Entity Type Airline

▼ Change Request

General Data Notes Attachments

Processing: You are
 Description: * testmh
 Priority:
 Created On: 23.11.20
 Changed On:
 Finalized On:

URL Selection

Client	Airline	URL
201	MH9	www.sap.com

OK Cancel

Main Data Attachments

Airline Details

Airline: * mh9
 Airline URL: * http://www.bmw.com
 Airline local currency: EUR

When you click on the 'OK' button the selected URL 'www.sap.com' in the popup is transferred to the field Airline URL on the UI.

Process Change Request: Airline MH9

Save Finalize Processing Send for Revision Close | Read Only | Check Show ▾ Run Validation Validation Log

☒ Airline MH9 URL overwritten

Change Request 625 Type Process Airline Status Changes to Be Executed Entity Type Airline

▼ Change Request

General Data Notes Attachments

Processing: You are the processor of the change request
 Description: * testmh1338
 Priority: ▾
 Created On: 23.11.2012 13:50:55
 Changed On:
 Finalized On:

Main Data Attachments


Airline Details

Airline: * mh9
 Airline URL: * http://www.sap.com
 Airline local currency: EUR

Remark: When you press the 'Cancel' button in the popup the value in field Airline URL won't be overwritten.

2 Create change request type

Call transaction **MDGIMG** and create a change request type e.g. **SFCMH01**.



Change View "Type of Change Request": Overview

New Entries BC Set: Change Field Values

Dialog Structure


- Type of Change Request
 - Entity Types
 - Business Activities
 - Service Level Agree

Type of Chg. Request	Edition Type	Data Model	Description (medium text)	Objects Required	Single Object	Main Entity Type	Workflow	Target Sys	ParallelCR
SFCMH01		SF	Process Airline	<input type="checkbox"/>	<input checked="" type="checkbox"/>	CARR	WS54300020	<input type="checkbox"/>	<input type="checkbox"/>
SFCQ_M		SF	Change Multiple Airlines for DQR	<input type="checkbox"/>	<input type="checkbox"/>	CARR	WS54300020	<input type="checkbox"/>	<input type="checkbox"/>
SFCQ_S		SF	Change Single Airline for DQR	<input type="checkbox"/>	<input type="checkbox"/>	CARR	WS54300020	<input type="checkbox"/>	<input type="checkbox"/>

In this example the change request type refers to the workflow **WS54300020**.

3 Assign processor to workflow step number

Now assign your user as a processor to each change request/workflow step combination.



Change View "Assignment of Processors to Workflow Step Number": Overview


New Entries BC Set: Change Field Values

Assignment of Processors to Workflow Step Number

Type of Chg. Request	S..	Description (medium)	O. Agent ID	Complete name
SFCMH01	0	Request	US HUBERTM	Matthias Hubert
SFCMH01	1	Processing	US HUBERTM	Matthias Hubert
SFCMH01	<input type="checkbox"/>	Final Check	US HUBERTM	Matthias Hubert

4 Create Enrichment Adapter Class

Go to transaction **SE24** in order to create the class which implements the interface **IF_USMD_ENRICHMENT_ADAPTER**.



Then implement the method **IF_USMD_ENRICHMENT_ADAPTER~EXECUTE**. The method contains the following signature.

Ty.	Parameter	Type spec.	Description
>>	C_LAST_USER_ACTION	TYPE MDG_SDQ_ENRICH_ACTION_CODE	Last user action
>>	ET_MESSAGE	TYPE USMD_T_MESSAGE	Messages
>>	EF_USER_ACTION_NECESSARY	TYPE USMD_FLG	FIN MDM: General characteristics
>>	ER_DATA	TYPE REF TO DATA	Data that will be used for user interaction
>>	ET_ADDITIONAL_DATA	TYPE USMD_TS_VALUE	Field Name and Value
>>	CS_DATA	TYPE REF TO DATA	Data to be enriched

Method	IF_USMD_ENRICHMENT_ADAPTER~EXECUTE	Active
--------	------------------------------------	--------

The method could be implemented as follows:

```
DATA: lt_urls TYPE STANDARD TABLE OF zmh_airline_url,  
      ls_urls type zmh_airline_url.
```

```
FIELD-SYMBOLS: <lt_ui_data>      TYPE STANDARD TABLE,  
               <ls_data>         type any.
```

```
* as in address enrichment  
CLEAR: et_additional_data,et_message.  
ASSIGN cs_data->* TO <ls_data>.
```

```
if c_last_user_action = 'ENRI'.  
* avoid that when 'OK' button in url popup is pressed that popup is shown a second time  
  ef_user_action_necessary = abap_false.  
  return.  
endif.
```

```
* read airline urls to be displayed in popup  
SELECT * FROM zmh_airline_url INTO TABLE lt_urls.
```


```
CREATE DATA er_data TYPE STANDARD TABLE OF zmh_airline_url.  
ASSIGN er_data->* TO <lt_ui_data>.  
<lt_ui_data> = lt_urls.
```

ef_user_action_necessary = abap_true.

Short description of the sample implementation:

In the implementation it is first checked whether the popup is already displayed. If it is displayed the flag **ef_user_action_necessary** is set to false to avoid that the popup is displayed again after pressing the 'OK'-button in the popup. In case the popup is not displayed airline URLs are read from database table **ZMH_AIRLINE_URL** and transferred to the exporting parameter **er_data**.


Definition of database table **ZMH_AIRLINE_URL**:



The screenshot shows the SAP Table Definition for **ZMH_AIRLINE_URL**. The table is active and has the short description 'Airline URL'. The 'Fields' tab is selected, showing the following fields:

Field	Key	Ini...	Data element	Data Type	Length	Deci...	Short Description
MANDT	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	MANDT	CLNT	3		0 Client
CARR	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	MDG S CARR ID	CHAR	3		0 Airline Code
URL	<input type="checkbox"/>	<input type="checkbox"/>	S_CARRURL	CHAR	255		0 Airline URL

Content of database table:




The screenshot shows the SAP Data Browser for the table **ZMH_AIRLINE_URL**. The title bar indicates 'Data Browser: Table ZMH_AIRLINE_URL Select Entries 1'. The table contains the following data:
















MANDT	CARR	URL
201	MH9	www.sap.com

5 Create Enrichment Feeder Class

Go to transaction **SE24** and create a class which implements the interface **IF_USMD_ENRICHMENT_FEEDER**.



Now implement the following methods:

Class Interface		ZCL_MH_ENRICH_FEEDER		Implemented / Active											
Properties		Interfaces		Friends		Attributes		Methods		Events		Types		Aliases	
<div><input type="checkbox"/> Parameter</div> <div> Exception</div> <div></div> <div></div> <div></div> <div></div> <div></div> <div></div> <div></div> <div></div> <div></div> <div></div> <div></div> <div></div> <div></div> <div></div> <div><input type="checkbox"/> Filter</div>															
Method		Level		Visibility		M.		Description							
IF_USMD_ENRICHMENT_FEEDER-GET_ADAPTER_DATA		Instanc_		Public				MDG data -> adapter specific data							
IF_USMD_ENRICHMENT_FEEDER-GET_MDG_DATA		Instanc_		Public				Adapter specific data -> MDG data							
IF_USMD_ENRICHMENT_FEEDER-PREPARE_UI_DATA		Instanc_		Public				Prepare UI data from adapter format to MDG format							
IF_USMD_ENRICHMENT_FEEDER-CONVERT_ACTION_RESULT		Instanc_		Public				Convert Action result to adapter format							
IF_USMD_ENRICHMENT_FEEDER-IS_RELEVANT		Instanc_		Public				Is the enrichment relevant for the given OTC code							
IF_USMD_ENRICHMENT_FEEDER-GET_RELEVANT_ENTITIES		Instanc_		Public				Get relevant entities to be read and to be updated							

Sample implementations for each of these methods are specified below:

Method IF_USMD_ENRICHMENT_FEEDER~GET_ADAPTER_DATA

method IF_USMD_ENRICHMENT_FEEDER~GET_ADAPTER_DATA.

```
data: ls_key_all type USMD_SX_ENTITY_DATA,
      ls_adapter_data type ZMH_AIRLINE_URL.
```

```
FIELD-SYMBOLS: <lt_key_all_data> type sorted TABLE,
                <ls_key_all_data> type any,
                <lv_key_all_data> type any,
                <lt_adapter_data> TYPE SORTED TABLE,
                <lv_adapter_data> type any.
```

```
* convert the data from MDG Format (=IT_KEY_ALL) to Adapter Format (=ER_DATA)
CREATE DATA er_data TYPE SORTED TABLE OF ZMH_AIRLINE_URL WITH UNIQUE KEY carr.
ASSIGN er_data->* TO <lt_adapter_data>.
```

```
read table it_key_all into ls_key_all
  with TABLE KEY entity = 'CARR'.
if sy-subrc = 0.
  ASSIGN ls_key_all-r_data->* TO <lt_key_all_data>.
  loop at <lt_key_all_data> ASSIGNING <ls_key_all_data>.
    ASSIGN COMPONENT 'CARR' OF STRUCTURE <ls_key_all_data> TO <lv_key_all_data>.
    ASSIGN COMPONENT 'CARR' OF STRUCTURE ls_adapter_data to <lv_adapter_data>.
    <lv_adapter_data> = <lv_key_all_data>.
    ASSIGN COMPONENT 'URL' OF STRUCTURE <ls_key_all_data> TO <lv_key_all_data>.
    ASSIGN COMPONENT 'URL' OF STRUCTURE ls_adapter_data to <lv_adapter_data>.
    <lv_adapter_data> = <lv_key_all_data>.
    ASSIGN COMPONENT 'MANDT' OF STRUCTURE ls_adapter_data to <lv_adapter_data>.
    <lv_adapter_data> = sy-mandt.
    insert ls_adapter_data into table <lt_adapter_data>.
  ENDLOOP.
endif.
endmethod.
```

Method IF_USMD_ENRICHMENT_FEEDER~GET_MDG_DATA

method IF_USMD_ENRICHMENT_FEEDER~GET_MDG_DATA.

```
data: lo_model type REF TO IF_USMD_MODEL_EXT,  
      lr_data  type ref to data,  
      lr_strdescr TYPE REF TO cl_abap_structdescr,  
      ls_component TYPE abap_componentdescr.
```

```
FIELD-SYMBOLS: <lt_data_write> type SORTED TABLE,  
               <ls_data_write> type any,  
               <lv_data_write> type any,  
               <ls_data>       type any,  
               <ls_data_db>    type zmh_airline_url,  
               <lv_data>       type any,  
               <lv_data_db>    type any.
```

```
CALL METHOD CL_USMD_FACTORY_EXT=>GET_MODEL_INSTANCE
```

```
EXPORTING
```

```
  i_usmd_model = 'SF'
```

```
IMPORTING
```

```
  eo_instance = lo_model.
```

```
CALL METHOD lo_model->CREATE_DATA_REFERENCE
```

```
EXPORTING
```

```
  i_fieldname = 'CARR'
```

```
  i_struct     = lo_model->GC_STRUCT_KEY_ATTR
```

```
  if_table     = 'X'      " Financial MDM: General Indicator
```

```
  i_tabtype    = lo_model->GC_TABTYPE_SORTED
```

```
IMPORTING
```

```
  er_data      = er_data.
```

```
assign er_data->* to <lt_data_write>.  
create data lr_data like line of <lt_data_write>.  
assign lr_data->* to <ls_data_write>.  
create data lr_data like line of it_data.  
assign lr_data->* to <ls_data>.
```

```
read table it_data ASSIGNING <ls_data>  
  index 1.
```

```
if sy-subrc <> 0.
```

```
  return.
```

```
endif.
```

```
ASSIGN COMPONENT 'WRITE_DATA' OF STRUCTURE <ls_data> TO <lv_data>.  
assign <lv_data>->* to <ls_data_db>.  
ASSIGN COMPONENT 'URL' OF STRUCTURE <ls_data_db> TO <lv_data_db>.  
ASSIGN COMPONENT 'URL' OF STRUCTURE <ls_data_write> TO <lv_data_write>.  
<lv_data_write> = <lv_data_db>.  
ASSIGN COMPONENT 'CARR' OF STRUCTURE <ls_data_db> TO <lv_data_db>.  
ASSIGN COMPONENT 'CARR' OF STRUCTURE <ls_data_write> TO <lv_data_write>.  
<lv_data_write> = <lv_data_db>.  
ASSIGN COMPONENT 'CARRNAME' OF STRUCTURE <ls_data_write> TO <lv_data_write>.  
<lv_data_write> = 'mh9'.  
ASSIGN COMPONENT 'CURRCODE' OF STRUCTURE <ls_data_write> TO <lv_data_write>.  
<lv_data_write> = 'EUR'.  
insert <ls_data_write> into table <lt_data_write>.  
endmethod.
```

Method IF_USMD_ENRICHMENT_FEEDER~PREPARE_UI_DATA

method IF_USMD_ENRICHMENT_FEEDER~PREPARE_UI_DATA.

```
  er_data = ir_data.
```

endmethod.

Method IF_USMD_ENRICHMENT_FEEDER~CONVERT_ACTION_RESULT

method IF_USMD_ENRICHMENT_FEEDER~CONVERT_ACTION_RESULT.

```
  er_data = ir_data.
```

endmethod.

Method IF_USMD_ENRICHMENT_FEEDER~IS_RELEVANT

```
method IF_USMD_ENRICHMENT_FEEDER~IS_RELEVANT.  
    ev_relevant = abap_true.  
endmethod.
```


Method IF_USMD_ENRICHMENT_FEEDER~GET_RELEVANT_ENTITIES

```
method IF_USMD_ENRICHMENT_FEEDER~GET_RELEVANT_ENTITIES.  
    data: lv_entity type USMD_ENTITY,  
          lv_attribute type USMD_FIELDNAME.
```


```
* entity 'Airline'  
    lv_entity = 'CARR'.  
    append lv_entity to et_read_entities.  
    e_write_entity = lv_entity.  
endmethod.
```

6 Create Popup wich displays a table of URLs


Create a WebDynpro Component e. g. **ZMH_AIRLINE_URL_SELECT** which implements WebDynpro Interface **MDG_ENRICHMENT_INTF**. Create a context node URLs in the Component Controller of the WebDynpro Component:




Then create a view **MAIN** and embed it into window **ZMH_AIRLINE_URL_SELECT**:



Map the node **URLS** of the Component Controller Context to the View Context:



Then use the wizard to define a table on View **MAIN** which displays the URLs from database table **ZMH_AIRLINE_URL** (see above).



Bind the table to the context node **URLS** (see above).

Now implement the following methods in View **MAIN**:

```
method WDDOMODIFYVIEW .
  IF first_time = abap_true.

    wd_comp_controller->go_popup->subscribe_to_button_event(
      button = if_wd_window=>co_button_ok
      button_text = 'OK'
      action_name = 'SUBMIT'
      action_view = view ).
```

```

        wd_comp_controller->go_popup->subscribe_to_button_event(
            button = if_wd_window=>co_button_cancel
            button_text = 'Cancel'
            action_name = 'CANCEL'
            action_view = view ).

ENDIF.
endmethod.

method ONACTIONSUBMIT .
    DATA lo_node_urls          TYPE REF TO if_wd_context_node.
    DATA lo_elem_urls          TYPE REF TO if_wd_context_element.
    DATA ls_urls                TYPE wd_this->ELEMENT_URLS.
    DATA lo_componentcontroller TYPE REF TO ig_componentcontroller.
    DATA lr_data                TYPE REF TO data.
    DATA lv_action              TYPE char4.
    data lv_url                  type S_CARRURL.

    FIELD-SYMBOLS: <ls_data>    TYPE zmh_airline_url.
    lo_node_urls = wd_context->get_child_node( name = wd_this->WDCTX_URLS ).

    * get element via lead selection
    lo_elem_urls = lo_node_urls->get_element( ).

    TRY.
        lo_elem_urls->get_static_attributes(
            IMPORTING
                static_attributes = ls_urls ).
        CATCH cx_sy_ref_is_initial.
    ENDTRY.

    IF ls_urls IS INITIAL.
        lv_action = 'CNCL'.
    ELSE.
        lv_action = 'ENRI'.
    ENDIF.

    CREATE DATA lr_data TYPE zmh_airline_url.
    ASSIGN lr_data->* TO <ls_data>.
    <ls_data> = ls_urls.


    lo_componentcontroller = wd_this->get_componentcontroller_ctr( ).

    lo_componentcontroller->fire_end_ui_interaction_evt(
        er_data = lr_data
        ev_action_code = lv_action
    ).

endmethod.

```

Then assign WebDynpro Interface **MDG_ENRICHMENT_INTF** to WebDynpro Component **ZMH_AIRLINE_URL_SELECT**.



Then implement the method `SHOW_ENRICHMENT_UI` of WebDynpro Interface `MDG_ENRICHMENT_INTF`.

A sample implementation of the method `SHOW_ENRICHMENT_UI` is specified here:

```
method SHOW_ENRICHMENT_UI .
  DATA lr_popup          TYPE REF TO if_wd_window.
  DATA lr_api_comp_controller TYPE REF TO if_wd_component.
  DATA lr_window_manager  TYPE REF TO if_wd_window_manager.
  DATA lo_wd_component    TYPE REF TO if_wd_component.
  DATA lo_window_manager  TYPE REF TO if_wd_window_manager.
  DATA lo_node_urls       TYPE REF TO if_wd_context_node.
  DATA lt_urls_elem       TYPE wd_this->Elements_urls.

  FIELD-SYMBOLS: <fs_data> TYPE any.
  ASSIGN ir_data->* TO <fs_data>.
  CHECK <fs_data> IS ASSIGNED.

  lt_urls_elem = <fs_data>.

  lo_node_urls = wd_context->get_child_node( name = wd_this->wdctx_urls ).
  lo_node_urls->bind_table( new_items = lt_urls_elem set_initial_elements = abap_true ).

  lr_api_comp_controller = wd_this->wd_get_api( ).
  lr_window_manager = lr_api_comp_controller->get_window_manager( ).


  lr_popup = lr_window_manager->create_window(
    modal          = abap_true
    window_name    = 'ZMH_AIRLINE_URL_SELE'
    title          = 'URL Selection'
    close_button   = abap_false
    button_kind    = if_wd_window=>co_buttons_okcancel
    message_type   = if_wd_window=>co_msg_type_none
    close_in_any_case = abap_true
    message_display_mode = if_wd_window=>co_msg_display_mode_none
  ).

  *Set the height and width here
  lr_popup->set_window_size( width = '40%' height = '5%' ).
  wd_this->GO_POPUP = lr_popup.

  lr_popup->open( ).
endmethod.
```

7 Define Enrichment Spot

Go to transaction **MDGIMG**.



Now enter the Enrichment Adapter class and the WebDynpro Component specified above.

Display View "Enrichment Adapters": Overview

Dialog Structure

- Enrichment Adapters
 - Enrichment Spot
 - Enrichment Adapter

Enrichment Adapter	UI Component	Description
ADDR_ENRICH_ADAPTER	MDG_SDQ_ADDR_ENRICHMENT	Address Enrichment Adapter
TAXJUR_ENRICH_ADAPTER	MDG_SDQ_TAXJUR_ENRICHMENT	Tax Jurisdiction Enrichment Adapter
ZMH_AIRL_URL_ENRICH_ADAPTER	ZMH_AIRLINE_URL_SELECT	Select an URL for an airline

Then define the enrichment spot which refers to the enrichment feeder class specified above.

Display View "Enrichment Spot": Overview

Dialog Structure

- Enrichment Adapters
 - Enrichment Spot
 - Enrichment Adapter

Enrichment Spot	Enrichment Feeder Class	Description
ENRICH_ADDRESS	CL_USMD_ADDR_ENRICH_FEEDER	Address Enrichment
ENRICH_TAXJUR	CL_USMD_ADDR_ENRICH_FEEDER	Tax Jurisdiction Enrichment
ZMH_AIRLINE_URL_SELECTION	ZCL_MH_ENRICH_FEEDER	Enrichment for airline url selection

Then assign the enrichment adapter class (see above) to the enrichment spot.

Display View "Enrichment Adapter Assignment": Overview

Dialog Structure


- Enrichment Adapters
 - Enrichment Spot
 - Enrichment Adapter

Enrichment Spot: ZMH_AIRLINE_URL_SELECTION


Enrichment Adapter
ZMH_AIRL_URL_ENRICH_ADAPTER

8 Configure Properties of Change Request Step


Go to transaction **MDGIMG**.



Then select the type of change request which was defined in one of the preceding steps (see above).



Then display the steps of the change request type.



Now maintain the enrichment spot for each of the steps. Select 'Relevant=true' and 'Execution=Always executed' to make sure the enrichment is executed. In the following screenshots step '00' is shown:

Display View "Enhancements and Checks per Change Request Step": Overview

Dialog Structure

- Type of Change Request
 - Change Request Step
 - Enhancements and Checks per Change Request Step
 - Entity Types per Change Request Step
 - Attributes per Change Request Step
 - User Interface per Change Request Step

Type of Chg. Request: SFCMH01
Chg. Req. Step: 00

Enhancements and Checks per Change Request Step				
Checks and Enrichment Spots	Message Output	Relevant	Execution	
Basic Check	Standard	<input checked="" type="checkbox"/>	Always executed	
Authorization Check	Standard	<input checked="" type="checkbox"/>	Always executed	
Duplicate Check	Standard	<input checked="" type="checkbox"/>	Executed when data changes	
Validation Rules (BRF+)	Standard	<input checked="" type="checkbox"/>	Always executed	
BAPI Validations	Standard	<input checked="" type="checkbox"/>	Always executed	
Existence Check	Standard	<input checked="" type="checkbox"/>	Always executed	
Reuse Area Check	Standard	<input checked="" type="checkbox"/>	Always executed	
Enrichment for airline url selection	Standard	<input checked="" type="checkbox"/>	Always executed	

Now all required settings are available, and you can execute the business scenarios specified at the beginning of this document. www.sap.com.

9 Additional Information

9.1 Further Reading

9.1.1 Information on SAP MDG on SAP S/4HANA

- Exchange knowledge: [SAP Community](#) | [Q&A](#) | [Blog](#)
- Try SAP Master Data Governance on S/4HANA for free: [Trial Version](#)
- Try SAP Master Data Governance on S/4HANA on the SAP Cloud Appliance Library: [S/4HANA 2022 FPS1](#)
- Learn more: [Latest Release](#) | [Help Portal](#) | [How-to Information](#) | [Key Presentations](#)

9.1.2 SAP Roadmap Explorer

- Please see the [roadmap for SAP Master Data Governance](#)

9.1.3 Related Information

- Learn more: [Floorplan Manager for Web Dynpro ABAP](#) | [How to Adapt FPM](#) | [FPM Blog](#) | [How-to Information](#) | [Service Mapping Tool](#) | [SAP S/4HANA Cookbook CVI](#)

9.2 SAP Notes

In addition to the detailed explanations written in this document, please see the following SAP Notes for further important information.

Note	Description
1619534	How to Create, Enhance and Adapt FPM Applications

